## Bases de données 1

## TD 5 - Algèbre relationnelle - Correction

#### Matthieu Nicolas

## Exercice 1 - Vente par Correspondance

Soit la base de données relationnelle décrite par le schéma relationnel suivant :

Client(noCli, nom, prenom, ddn, rue, CP, ville)

Fournisseur(noFour, raisonSoc)

Produit(noProd, libelle, pu, noFour)

Commande(noCom, noCli, date)

PartieCommande(noPartCom, noCom, noProd, qte)

Construire les requêtes algébriques répondant aux questions suivantes :

1. Libellé et prix unitaire de tous les produits

2. Libellé des produits de prix inférieur à 50€

```
R1 = RESTRICT(Produit, pu < 50)
Res = PROJECT(R1, libelle, pu)
```

3. Nom et prénom des clients ayant effectué une commande le 2020-09-17

```
R1 = RESTRICT(Commande, date = "2020-09-17")
R2 = JOIN(Client, R1, Client.noCli = R1.noCli)
Res = PROJECT(R2, nom, prenom)
```

4. Nom et prénom des clients ayant commandé au moins le produit p56

```
R1 = RESTRICT(PartieCommande, noProd = "p56")
R2 = JOIN(Commande, R1, Commande.noCom = R1.noCom)
R3 = JOIN(Client, R2, Client.noCli = R2.noCli)
Res = PROJECT(R3, nom, prenom)
```

5. Nom des clients ayant commandé au moins un produit de prix supérieur à  $500 \epsilon$ 

```
R1 = RESTRICT(Produit, pu > 500)
R2 = JOIN(PartieCommande, R1, PartieCommande.noProd = R1.noProd)
R3 = JOIN(Commande, R2, Commande.noCom = R2.noCom)
R4 = JOIN(Client, R3, Client.noCli = R3.noCli)
Res = PROJECT(R4, nom)
```

6. Nom des clients n'ayant pas commandé le produit p56

```
R1 = RESTRICT(PartieCommande, noProd = "p56")
R2 = JOIN(Commande, R1, Commande.noCom = R1.noCom)
R3 = JOIN(Client, R2, Client.noCli = R2.noCli)
Res = MINUS(PROJECT(Client, nom), PROJECT(R3, nom))
```

# Exercice 2 - Bibliothèque

Soit la base de données relationnelle décrite par le schéma relationnel suivant :

Ouvrage (noOuvrage, titre, auteur, editeur, dateSortie)

MotCle(noMotCle, libelle)

**Description**(noOuvrage, noMotCle)

**Exemplaire**(noExemplaire, position, dateAchat, noOuvrage)

Lecteur(noLecteur, nom, adresse)

Emprunt (no Emprunt, date Debut, date Retour, no Lecteur)

PartieEmprunt(noEmprunt, noExemplaire)

Construire les requêtes algébriques répondant aux questions suivantes :

1. Titre et auteur de tous les livres sortis en 2020

2. Titre et auteur de tous les livres sortis en 2020 dont la bibliothèque possède au moins un exemplaire

3. Titre et auteur de tous les livres dont la bibliothèque possède au moins un exemplaire et qui possèdent comme mots-clés "Fantastique" et "Horreur"

```
R1 = JOIN(Ouvrage, Exemplaire, Ouvrage.noOuvrage = Exemplaire.noExemplaire)
R2 = JOIN(MotCle, Description, MotCle.noMotCle = Description.noMotCle)
R3 = RESTRICT(R2, Libelle = "Horreur")
R4 = RESTRICT(R2, Libelle = "Fantastique")
R5 = JOIN(R1, R3, R1.noOuvrage = R3.noOuvrage)
R6 = JOIN(R5, R4, R5.noOuvrage = R4.noOuvrage)
Res = PROJECT(R6, titre, auteur)
```

4. Nom des lecteurs ayant un emprunt en cours à la date du 2020-09-29

5. Nom des lecteurs ayant emprunté un livre écrit par Robin Hobb

```
R1 = RESTRICT(Ouvrage, auteur = "Robin Hobb")
R2 = JOIN(Exemplaire, R1, Exemplaire.noOuvrage = R1.noOuvrage)
R3 = JOIN(PartieEmprunt, R2, PartieEmprunt.noExemplaire = R2.noExemplaire)
R4 = JOIN(Emprunt, R3, Emprunt.noEmprunt = R3.noEmprunt)
R5 = JOIN(Lecteur, R4, Lecteur.noLecteur = R4.noLecteur)
RES = PROJECT(R5, nom)
```